
Introduction
Blowfish is a patent-free and royalty-free encryption and decryption scheme. It was invented by Bruce Schneier in 1994 and 
has not been broken yet (at least successful attempts have not been published). If you want in-depth information on the 
algorithm, please visit http://www.schneier.com/blowfish.html

Blowfish is a symmetrical cipher, meaning that you use the same key for encryption and decryption. The key can have 
up to 56 extended ASCII characters (using the full ASCII range of codes 0 … 255). The fact that Blowfish is symmetrical 
means that you have to guard closely the key you are using for decryption and encryption.

This library for the hp49g+ pocket calculator implements Blowfish key initialization, the encoding of any amount of 
text (in a string) into a list of codes, and the decoding of such a list back into the original plain text. In addition, one rou-
tine is supplied that, when called, resets the subkey lists to some arbitrary initial value (the digits of π) so that no-one can 
use your subkeys to encode or decode messages. It is important that you call the reset procedure when you’re done with 
encoding and decoding! I don’t know whether the library works on hp pocket calculators other than the hp49g+.

The initialization (the core of which is written in SysRPL) takes quite a bit of time (around 6 minutes) but needs to be 
called only when you change your key. After that, you can encode text and decode relatively quickly using the precalculated 
subkey lists.

Installation
Transfer the binary directory file BF06.hp (indicating version 0.5) to your hp49g+ calculator to into any directory you 
please. If you don’t have the BF06.hp file, you can extract two versions (one binary and one text) from inside this pdf using 
the Adobe Acrobat application. Now, inside the freshly transferred directory, you will find, amongst files named »Sbox«, 
»S0«, »S1«, »S2«, »S3«, and »Pbox«, the following six files:

Usage
»BFini« is the first thing you execute in order to initialize Blowfish for use with a particular key. Put your key as a string 

on the stack (it may be as long as you like although only the first 56 characters will be used as to not weaken the algorithm), 
press »BFini«, wait… wait… and finally the subkeys have been prepared for encoding and decoding. While the current 
initialization (6 minutes), written in SysRPL, features a great increase in speed over the original UserRPL version (29.5 
minutes), it definitely could use more speed. I hope to be able to provide a faster version in the near future. Example: 
"MIchAEl" BFini

»BFenc« is the main encoding routine. Put the text you want to encode on the stack (level 1) and press »BFenc«. After 
a few seconds, the stack will show a list of hex string values. This represents the encoded message. You can transmit this 
sequence to another hp Blowfish user (who must know the key), and s/he can then decode the original message. Provi-
ded you have not changed the subkeys in the meantime, to decode, put the list of codes on the stack and press »BFdec«. 
Should you have pressed »Reset« in the meantime or encoded messages with other keys, you must re-initialize with the key 
and »BFini«, then put the codelist on level 1 and press »BFdec«. Returned will be the original message plus possible space 
padding bytes (which may have been appended in order to pad the string up to a multiple of 8 bytes).

Press »Reset« to prevent anyone from using your subkeys for encoding or decoding your messages. After Reset, when 
you want to encode or decode more messages, you must re-initialize with a key and »BFini«.

The »BFe6s« command may be used by more experienced hp users. It expects two hex strings (of at most 32 bits each) 
on the stack and encodes them using the Blowfish algorithm into another pair of hex strings on the stack. »BFd6s« decodes 
such a pair, using the same subkeys. Please note: after encoding, you must not execute »Reset« because that would scramble 
your subkeys and make decoding impossible. If you should happen to press »Reset« before decoding, you must re-initialize 
the subkeys again with »BFini« and the original key.

Following are the source listings for the mentioned modules and for the function f2 used in the algorithm.

http://www.schneier.com/blowfish.html


Blowfish on the hp49g+

%%HP: T(3)A(R)F(.);
\<< 0 0 \-> K Z MA
  \<< HEX
      Pbox 'P' STO
      Sbox 1 GET 'S0' STO
      Sbox 2 GET 'S1' STO
      Sbox 3 GET 'S2' STO
      Sbox 4 GET 'S3' STO

      1 'Z' STO
      K SIZE 56 MIN 'MA' STO

      1 18
      FOR I
        # 0d

        0 3
        FOR R
          SLB
          # 4294967295d AND
          K Z Z SUB
          NUM R\->B
          OR
          Z 1 +
          DUP
          MA > IF THEN DROP 1 END
          'Z' STO
        NEXT
        P I GET XOR
        P SWAP I SWAP PUT 'P' STO
      NEXT

      # 0d # 0d

      1 17
      FOR I
        BFe6s DUP2 P SWAP I 1 + SWAP PUT
        SWAP I SWAP PUT 'P' STO
      2 STEP

      1 255
      FOR J
        BFe6s DUP2 S0 SWAP J 1 + SWAP PUT
        SWAP J SWAP PUT 'S0' STO
      2 STEP

      1 255
      FOR J
        BFe6s DUP2 S1 SWAP J 1 + SWAP PUT
        SWAP J SWAP PUT 'S1' STO
      2 STEP

      1 255
      FOR J
        BFe6s DUP2 S2 SWAP J 1 + SWAP PUT
        SWAP J SWAP PUT 'S2' STO
      2 STEP

      1 255
      FOR J
        BFe6s DUP2 S3 SWAP J 1 + SWAP PUT
        SWAP J SWAP PUT 'S3' STO
      2 STEP

      DEC

      DROP2
  \>>
\>>

    hp prologue, flag settings
    the key is stored in 'K'
    set hexadecimal mode
    Reinstate the original P-box entries
    … and the original 4 S-boxes

         
     
        
    There are 18 P-box entries to be
    scrambled by logical ORing with
    characters from the user key.

    The first #0d is the left 32 bits
    and the second the right 32 bits of
    data
    
    Now the P-box gets Blowfish-encoded

    
    And now the S-boxes

    Restore decimal operating mode

    and drop the intermediate data

Description: this is the UserRPL function BFini. It scrambles the subkeys using the entries from the original 
S-boxes (the digits of Pi minus the leading »3«) and the P-box entries using the key, too.
Call it to make the current key the working key.

 Stack: (key — )



Blowfish on the hp49g+

%%HP: T(3)A(R)F(.);
\<< 0 \-> T RO
  \<< HEX
      T SIZE 8 IDIV2
      DUP

      0 ==
      IF THEN DROP
      ELSE T SWAP 8 SWAP - 1 SWAP
           FOR I " " + NEXT
      'T' STO
      1 +
      END
  
      'RO' STO

      { }

      1 RO
      FOR I
        T DUP HEAD NUM 16777216 *
        SWAP TAIL DUP HEAD NUM 65536 * ROT
        +
        SWAP TAIL DUP HEAD NUM 256 * ROT
        +
        SWAP TAIL DUP HEAD NUM
        ROT
        +
        R\->B
        SWAP TAIL DUP HEAD NUM 16777216 *
        SWAP TAIL DUP HEAD NUM 65536 * ROT
        +
        SWAP TAIL DUP HEAD NUM 256 * ROT
        +
        SWAP TAIL DUP HEAD NUM ROT
        +

        SWAP

        TAIL 'T' STO
        R\->B

        BFe6s

        2
        \->LIST
        +
      NEXT
  \>>
\>>

    hp prologue, flag settings
    the plain text is stored in 'T'
    set hexadecimal mode
    Pad plaintext T if necessary
    so it has a length of a multiple of
    8 bytes afterwards. The 8 bytes are
    necessary as the basic unit of data
    encryption by the Blowfish algorithm
    as it encodes a pair of 32-bit values
    simultaneously.
    Save modified and padded text
     
     
        
    'RO' means encryption ROunds
     
    this empty list collects the codes
     
    Do 'RO' rounds of encryption
     
    Calculate a code for the next
    4 bytes of the plaintext
    (using their ASCII values)
     
    

    and access the next 4 characters
    and do the same for them
     
     
     

    get the order right: left, right    
     
    store the rest of the characters
    for later processing
    
    call the 64-bit encryption routine

    make a 2-element mini-list from them
    
    and put it into the collection list.

Description: this is the UserRPL function BFenc.
Using the subkeys in the four »s-boxes«, any string is encoded using the Blowfish algorithm.

 Stack: (text as string — list of hex-string codes)



Blowfish on the hp49g+

%%HP: T(3)A(R)F(.);
\<< "" \-> C T
  \<< HEX

      1 C SIZE 1 -
      FOR I
        C I GET
        C I 1 + GET

        BFd6s

        B\->R
        SWAP
        B\->R

        16777216 IDIV2
        SWAP CHR T SWAP +
        'T' STO
        65536 IDIV2 SWAP
        CHR T SWAP +
        'T' STO
        256 IDIV2 SWAP
        CHR T SWAP +
        'T' STO
        CHR T SWAP +
        'T' STO

        16777216 IDIV2
        SWAP CHR T SWAP +
        'T' STO
        65536 IDIV2 SWAP
        CHR T SWAP +
        'T' STO
        256 IDIV2 SWAP
        CHR T SWAP +
        'T' STO
        CHR T SWAP +
        'T' STO
     2
    STEP

    T
  \>>
\>>

    hp prologue, flag settings
    the code list is stored in 'C'
    the plain text in 'T'. 

    Run through code list C
    in steps of 2…
    retrieve first 32 bits of code
    retrieve second 32 bits of code
     
    call the 64-bit decoding routine
     
    Transform the result into real
    numbers for further arithmetic
    processing
        
    Split up the big numbers into 4
    single bytes, translate the bytes
    into character codes and store
    the growing plaintext in 'T'.
     
    

    Access the next 4 characters
    and do the same for them
     
     
     

    now access the next two codes    
     
     
    leave the plaintext on the stack
    as the result

Description: this is the UserRPL function BFdec.
Using the subkeys in the four »s-boxes«, any list of hex string codes is decrypted using the Blow-
fish algorithm.

 Stack: ( list of hex string codes — plaintext as string)



Blowfish on the hp49g+

%%HP: T(3)A(R)F(.);
\<< Pbox 'P' STO
    Sbox 1 GET ‚S0‘ STO
    Sbox 2 GET ‚S1‘ STO
    Sbox 3 GET ‚S2‘ STO
    Sbox 4 GET ‚S3‘ STO

    "Blowfish reset."
\>>

    hp prologue, flag settings
    Restore the P-box
    and the 4 S-boxes.

Description: this is the UserRPL function Reset.
To prevent tampering with the key-dependent subkeys, call this routine to restore the S-boxes 
and P-box entries to factory settings. DROP the comment afterwards.

 Stack: ( — "Blowfish reset.")



Blowfish on the hp49g+

%%HP: T(3)A(R)F(.);
"!RPL
!NO CODE
::
 CK2NOLASTWD
 CK&DISPATCH0
   #BB ::
       
    BINT17 BINT1
    DO
      SWAP
      ID P INDEX@ NTHCOMPDROP
      bitXOR
      SWAP
      OVER
      ID F2
      bitXOR
      SWAP
    LOOP 

    SWAP
    ID P BINT17 NTHCOMPDROP
    bitXOR
    
    SWAP
    ID P BINT18 NTHCOMPDROP
    bitXOR
    
    SWAP    
    ; 
;
@"

    hp prologue, flag settings
    This is a System RPL program
    Don’t produce code prologue for this
    Start secondary
    Error if no two arguments on stack
    Supported number types on stack are…
    HXS in level 1 and level 2
    
    Start a LOOP from 1 to 16
    
    Swap numbers on stack, data_l, data_r
    Get P-box element at loop index
    XOR with other data element
    Swap both 32-bit data components
    Process the upper one
    Call function f
    XOR with other data element
    and swap again
    
    
    Make the last swap undone
    access element P[17] of the P-box
    XOR 32-bit data element with data el.
    
    Swap data_l and data_r once more
    access element P[18] of the P-box
    XOR 32-bit data element with data el.
   
    restore correct order data_l’ data_r’

Description: this is the system-rpl function BFe6s. This routine encodes a pair of HXS 32-bit values on the 
stack, data_l and data_r. It returns two scrambled values on the stack, the Blowfish-encoded 
data_l’ and data_r’.  data_l are the left 32-bits of a 64-bit datum, data_r the right 32 bits.

 Stack: (data_l    data_r  —  data_l’    data_r’) (all values HXS hex strings)



Blowfish on the hp49g+

%%HP: T(3)A(R)F(.);
!RPL
!NO CODE
::
 CK2NOLASTWD
 CK&DISPATCH0
   #BB ::

    BINT19 BINT3
    DO
      SWAP
      ID P BINT21 INDEX@  #- NTHCOMPDROP
      bitXOR
      SWAP
      OVER
      ID F2
      bitXOR
      SWAP
    LOOP 

    SWAP
    ID P BINT2 NTHCOMPDROP
    bitXOR
    
    SWAP
    ID P BINT1 NTHCOMPDROP
    bitXOR
    
    SWAP    
    ; 
;
@

    hp prologue, flag settings
    System RPL program
    Produce no code prolog
 
    Two arguments on the stack required
    They must be of type
    Hex string and hex string

    Loop 16 times, from 18 downto 3
     
    exhange data_l’ and data_r’
    access P-box element 21–loop index
    and XOR P element with data_l’
    swap XOR(P[i], data_l’) with data_r’
    duplicate the XOR result below
    calculate f2(XOR result)
    XOR with data_r’
    restore stack order: data_l’, data_r’

    undo last swap of data_l’ & data_r’
    access P[2] of the P-box
    XOR 32-bit data element with P[2]
    
    and swap once more data_l’, data_r’
    access P[1] of the P-box
    XOR 32-bit data element with P[1]
    
    restore correct order data_l data_r

Description: this is the user-rpl function BFd6s. This routine decodes a pair of 32-bit values on the stack, 
data_l’ and data_r’. It returns two cleartext values on the stack, the Blowfish-decoded data_l and 
data_r.  data_l’ are the left 32-bits of a 64-bit datum, data_r’ the right 32 bits.

 Stack: (HXS data_l’    HXS data_r’  —  HXS data_l    HXS data_r)



Blowfish on the hp49g+

%%HP: T(3)A(R)F(.);
!RPL
!NO CODE
::
  CK1NOLASTWD
  DUP
  HXS 2 FF
  bitAND
  
  SWAP
  bitSRB
  DUP
  HXS 2 FF
  bitAND
  
  SWAP
  bitSRB
  DUP
  HXS 2 FF
  bitAND
  
  SWAP
  bitSRB
  HXS 2 FF
  bitAND

  HXS>#
  #1+
  ID S0
  SWAP
  NTHCOMPDROP
  
  SWAP
  HXS>#
  #1+
  ID S1
  SWAP
  NTHCOMPDROP
  
  bit+
  HXS 8 FFFFFFFF
  bitAND
  
  SWAP
  HXS>#
  #1+
  ID S2
  SWAP
  NTHCOMPDROP 
 
  bitXOR
  
  SWAP
  HXS>#
  #1+
  ID S3
  SWAP
  NTHCOMPDROP
  
  bit+
  HXS 8 FFFFFFFF
  bitAND
  ;
@

    hp prologue, flag settings
    system rpl language
    not a code object
    
    Is there any argument?
    duplicate for successive splitting
    keep only the lowest 8 bits
    by masking out any other bits: a

    get next-highest byte
    extract the lowest byte only
    duplicate for successive splitting
    keep only the lowest 8 bits
    by masking out any other bits: b

    get next-highest byte
    extract the lowest byte only
    duplicate for successive splitting
    keep only the lowest 8 bits
    by masking out any other bits: c

    get next-highest byte
    extract the lowest byte only
    keep only the lowest 8 bits
    by masking out any other bits: d

    make the hex string into a bint
    add 1 to the index a (list 1-based)
    ready for access the first s-box
    get access index into place
    and retrieve s-box element

    get the next index (b)    
    make the hex string into a bint
    add 1 to the index (list 1-based)
    ready for access the first s-box
    get access index into place
    and retrieve s-box element

    add: stack now holds S0[a]+S1[b]
    keep only the lowest 32 bits
    by masking out any other bits
 
    get the next index (c) 
    make it a bint number for indexing
    add 1 to the index (list 1-based)
    ready for access the first s-box
    get access index into place
    and retrieve s-box element

    XOR: stack now (S0[a]+S1[b])^S2[c]
     
    get the last index (d)
    make it a bint number for indexing
    add 1 to the index (list 1-based)
    ready for access the first s-box
    get access index into place
    and retrieve s-box element

    add: (S0[a]+S1[b])^S2[c] + S3[d]
    keep only the lowest 32 bits
    by masking out any other bits

Description: this is the system-rpl function f2 at the core of the Blowfish algorithm. From the subkeys in the 
four »s-boxes«, a kind of hash is calculated which is later used to scramble the left 32 bit of the 
data to be encoded and the right 32 bits. This version is over 3 x as fast as the User RPL version.

 Stack: (32-bit HXS z — HXS f(z))



%%HP: T(3)A(R)F(.);
DIR
  BFini
  \<< 0 0 \-> K Z MA
    \<< HEX Pbox 'P' STO Sbox 1 GET 'S0' STO Sbox 2 GET 'S1' STO Sbox 3 GET 'S2' STO Sbox 4 GET 'S3' STO 1 'Z' STO K SIZE 56 MIN 'MA' STO 1 18
      FOR I # 0h 0 3
        FOR R SLB # FFFFFFFFh AND K Z Z SUB NUM R\->B OR Z 1 + DUP MA >
          IF
          THEN DROP 1
          END 'Z' STO
        NEXT P I GET XOR P SWAP I SWAP PUT 'P' STO
      NEXT # 0h # 0h 1 17
      FOR I BFe6s DUP2 P SWAP I 1 + SWAP PUT SWAP I SWAP PUT 'P' STO 2
      STEP 1 255
      FOR J BFe6s DUP2 S0 SWAP J 1 + SWAP PUT SWAP J SWAP PUT 'S0' STO 2
      STEP 1 255
      FOR J BFe6s DUP2 S1 SWAP J 1 + SWAP PUT SWAP J SWAP PUT 'S1' STO 2
      STEP 1 255
      FOR J BFe6s DUP2 S2 SWAP J 1 + SWAP PUT SWAP J SWAP PUT 'S2' STO 2
      STEP 1 255
      FOR J BFe6s DUP2 S3 SWAP J 1 + SWAP PUT SWAP J SWAP PUT 'S3' STO 2
      STEP DEC DROP2
    \>>
  \>>
  BFenc
  \<< 0 \-> T RO
    \<< HEX T SIZE 8 IDIV2 DUP 0 ==
      IF
      THEN DROP
      ELSE T SWAP 8 SWAP - 1 SWAP
        FOR I " " +
        NEXT 'T' STO 1 +
      END 'RO' STO { } 1 RO
      FOR I T DUP HEAD NUM 16777216 * SWAP TAIL DUP HEAD NUM 65536 * ROT + SWAP TAIL DUP HEAD NUM 256 * ROT + SWAP TAIL DUP HEAD NUM ROT + R\->B SWAP TAIL DUP HEAD NUM 16777216 * SWAP TAIL DUP HEAD NUM 65536 * ROT + SWAP TAIL DUP HEAD NUM 256 * ROT + SWAP TAIL DUP HEAD NUM ROT + SWAP TAIL 'T' STO R\->B BFe6s 2 \->LIST +
      NEXT
    \>>
  \>>
  BFdec
  \<< "" \-> C T
    \<< HEX 1 C SIZE 1 -
      FOR I C I GET C I 1 + GET BFd6s B\->R SWAP B\->R 16777216 IDIV2 SWAP CHR T SWAP + 'T' STO 65536 IDIV2 SWAP CHR T SWAP + 'T' STO 256 IDIV2 SWAP CHR T SWAP + 'T' STO CHR T SWAP + 'T' STO 16777216 IDIV2 SWAP CHR T SWAP + 'T' STO 65536 IDIV2 SWAP CHR T SWAP + 'T' STO 256 IDIV2 SWAP CHR T SWAP + 'T' STO CHR T SWAP + 'T' STO 2
      STEP T
    \>>
  \>>
  Reset
  \<< Pbox 'P' STO Sbox 1 GET 'S0' STO Sbox 2 GET 'S1' STO Sbox 3 GET 'S2' STO Sbox 4 GET 'S3' STO "Blowfish reset."
  \>>
  BFe6s External External \164 BBh \164 11h \164 1h External External P External External External External External External F2 External External External External P \164 11h External External External External P \164 12h External External External External
  BFd6s External External \164 BBh \164 13h \164 3h External External P \164 15h External External External External External External External F2 External External External External P \164 2h External External External External P \164 1h External External External External
  F2 External External # FFh External External External External # FFh External External External External # FFh External External External # FFh External External External S0 External External External External External External S1 External External External External # FFFFFFFFh External External External External S2 External External External External External External External S3 External External External External # FFFFFFFFh External

